Introduction to VHDL

1

Course Objectives

- Learn the Basic Constructs of VHDL
- Learn the Modeling Structure of VHDL
- Understand the Design Environments
  - Simulation
  - Synthesis

2

Course Outline

- Introduction to Altera Devices & Altera Design Software
- VHDL Basics
  - Overview of Language
- Design Units
  - Entity
  - Architecture
  - Configurations
  - Packages (Libraries)
- Architecture Modeling Fundamentals
  - Signals
  - Processes
    - Sequential Statements

3

Course Outline

- Understanding VHDL and Logic Synthesis
  - Process Statement
  - Inferring Logic
- Model Application
  - State Machine Coding
- Hierarchical Designing
  - Overview
  - Structural Modeling
  - Application of Library of Parameterized Modules (LPMs)

4

Introduction to Altera Devices & Design Software

5

The Programmable Solutions Company®

- Programmable Devices
- Design Software
- Intellectual Property (IP)

6

Introduction to Altera Devices

- Programmable Logic Families
  - High & Medium Density FPGAs
    - Stratix™ II, Stratix, APEX™ II, APEX 20K, & FLEX® 10K
  - Low-Cost FPGAs
    - Cyclone™ & ACEX® 1K
  - FPGAs with Clock Data Recovery
    - Stratix GX & Mercury™
  - CPLDs
    - MAX® 7000 & MAX 3000
  - Embedded Processor Solutions
    - Nios™, Excalibur™
  - Configuration Devices
    - EPC

7

Introduction to Altera Design Software

- Software & Development Tools:
  - Quartus II
    - Stratix II, Stratix, Stratix GX, Cyclone, APEX II, APEX 20K/E/C, Excalibur, & Mercury Devices
    - FLEX 10K/A/E, ACEX 1K, FLEX 6000, MAX 7000S/AE/B, MAX 3000A Devices
  - Quartus II Web Edition
    - Free Version
    - Not All Features & Devices Included
  - MAX+PLUS® II
    - All FLEX, ACEX, & MAX Devices

8

Intellectual Property MegaStore™

9

VHDL Basics

10

VHDL

VHSIC (Very High Speed Integrated Circuit)

Hardware

Description

Language

11

What is VHDL?

- IEEE Industry Standard Hardware Description Language
- High-level Description Language for Both Simulation & Synthesis

12

VHDL History

- 1980 - U.S. Department of Defense (DOD) Funded a Project to Create a Standard Hardware Description Language Under the Very High Speed Integrated Circuit (VHSIC) Program
- 1987 - the Institute of Electrical and Electronics Engineers (IEEE) Ratified As IEEE Standard 1076
- 1993 - the VHDL Language Was Revised and Updated to IEEE 1076 '93

13

Terminology

- HDL - Hardware Description Language Is a Software Programming Language That Is Used to Model a Piece of Hardware
- Behavior Modeling - A Component Is Described by Its Input/Output Response
- Structural Modeling - A Component Is Described by Interconnecting Lower-level Components/Primitives

14

Behavior Modeling

- Only the Functionality of the Circuit, No Structure
- No Specific Hardware Intent
- For the Purpose of Synthesis, As Well As *Simulation*



Left Bit Shifter

15

Structural Modeling

- Functionality and Structure of the Circuit
- Call Out the Specific Hardware
- For the Purpose of Synthesis



16

More Terminology

- Register Transfer Level (RTL) - A Type of Behavioral Modeling, for the Purpose of Synthesis
  - Hardware Is Implied or Inferred
  - Synthesizable
- Synthesis - Translating HDL to a Circuit and Then Optimizing the Represented Circuit
- RTL Synthesis - The Process of Translating a RTL Model of Hardware Into an Optimized Technology Specific Gate Level Implementation

17

RTL Synthesis

```
Process (a, b, c, d, sel)
 begin
  case (sel) is
        when "00" => mux_out <= a;
        when "01" => mux_out <= b;
        when "10" => mux_out <= c;
        when "11" => mux_out <= d;
  end case;
```





18

VHDL Synthesis Vs. Other HDL Standards

- VHDL
  - "Tell Me How Your Circuit Should Behave and I Will Give You Hardware That Does the Job."
- Verilog
  - Similar to VHDL
- ABEL, PALASM, AHDL
  - "Tell Me What Hardware You Want and I Will Give It to You"

19

VHDL Synthesis vs. Other HDL Standards

- VHDL
  - "Give me a circuit whose output only changes when there is a low-to-high transition on a particular input. When the transition happens, make the output equal to the input until the next transition."
  - Result: VHDL Synthesis provides a positive edge-triggered flipflop
- ABEL, PALASM, AHDL
  - "Give me a D-type flipflop."
  - Result: ABEL, PALASM, AHDL synthesis provides a D-type flipflop. The sense of the clock depends on the synthesis tool.

20

Typical Synthesis Design Flow



21

Typical Simulation Design Flow



22

VHDL Basics

- Two Sets of Constructs:
  - Synthesis
  - Simulation
- The VHDL Language Is Made up of Reserved Keywords
- The Language Is, for the Most Part, **Not** Case Sensitive
- VHDL Statements Are Terminated With a **;**
- VHDL Is White Space Insensitive. Used for Readability.
- Comments in VHDL Begin With "**--**" to Eol
- VHDL Models Can Be Written:
  - Behavioral
  - Structural
  - Mixed

23

VHDL Design Units

24

VHDL Basics

- VHDL Design Units
  - Entity
    - Used to Define External View of a Model. I.E. Symbol
  - Architecture
    - Used to Define the Function of the Model. I.E. Schematic
  - Configuration
    - Used to Associate an Architecture With an Entity
  - Package
    - Collection of Information That Can Be Referenced by VHDL Models. I.E. Library
    - Consist of Two Parts Package Declaration and Package Body

25

Entity Declaration

```
ENTITY <entity_name> IS
        Generic Declarations
        Port Declarations
END <entity_name>; (1076-1987 version)
END ENTITY <entity_name> ; ( 1076-1993 version)
```

- Analogy : Symbol
- <Entity_name> Can Be Any Alpha/Numerical Name
  - Note: MAX+PLUS II Requires That the <Entity_name> and <File_name> Be the Same; Not Necessary in Quartus II
- Generic Declarations
  - Used to Pass Information Into a Model
  - Quartus II & MAX+PLUS II Place Some Restriction on the Use of Generics
- Port Declarations
  - Used to Describe the Inputs and Outputs i.e. Pins

26

Entity : Generic Declaration

```
ENTITY <entity_name> IS
        Generic ( constant tplh , tphl : time := 5 ns;
                  -- Note constant is assumed and is not required
                   tphz, tplz : time := 3 ns;
                   default_value : integer := 1;
                   cnt_dir : string := "up"
                 );
        Port Declarations
END <entity_name>; (1076-1987 version)
END ENTITY <entity_name> ; ( 1076-1993 version)
```

- New Values Can Be Passed During Compilation
- During Simulation/Synthesis a Generic Is Read Only

27

Entity : Port Declarations

```
ENTITY <entity_name> IS
        Generic Declarations
        Port ( signal clk : in bit;
                --Note: signal is assumed and is not required
                        q : out bit
             );
END <entity_name>; (1076-1987 version)
END ENTITY <entity_name> ; ( 1076-1993 version)
```

- Structure : <Class> Object_name : <Mode> <Type> ;
  - <Class> : What Can Be Done to an Object
  - Object_name : Identifier
  - <Mode> : Directional
    - **in** (Input) **Out** (Output)
    - **Inout** (Bidirectional) **Buffer** (Output W/ Internal Feedback)
  - <Type> : What Can Be Contained in the Object

Architecture

- Analogy : Schematic
- Describes the Functionality and Timing of a Model
- Must Be Associated With an **ENTITY**
- **ENTITY** Can Have Multiple Architectures
- Architecture Statements Execute Concurrently (Processes)
- Architecture Styles
  - Behavioral : How Designs Operate
    - RTL : Designs Are Described in Terms of Registers
    - Functional : No Timing
  - Structural : Netlist
    - Gate/Component Level
  - Hybrid : Mixture of the Above

Architecture

```
ARCHITECTURE <Identifier> OF <Entity_identifier> IS
--Architecture Declaration Section (List Does Not Include All)
    SIGNAL Temp : Integer := 1;  -- Signal Declarations :=1 Is Default Value Optional
    CONSTANT Load : Boolean := True;  --Constant Declarations
    TYPE States IS ( S1, S2, S3, S4) ;  --Type Declarations
    --Component Declarations Discussed Later
    --Subtype Declarations
    --Attribute Declarations
    --Attribute Specifications
    --Subprogram Declarations
    --Subprogram Body
BEGIN
    Process Statements
    Concurrent Procedural Calls
    Concurrent Signal Assignment
    Component Instantiation Statements
    Generate Statements
END <Architecture Identifier> ; (1076-1987 Version)
End ARCHITECTURE; (1076-1993 Version)
```

VHDL - Basic Modeling Structure

```
ENTITY entity_name IS
        generics
        port declarations
END entity_name;

ARCHITECTURE arch_name OF entity_name IS
        enumerated data types
        internal signal declarations
        component declarations
BEGIN
        signal assignment statements
        process statements
        component instantiations
END arch_name;
```


31

VHDL : Entity - Architecture



32

Configuration

- Used to Make Associations Within Models
  - Associate a Entity and Architecture
  - Associate a Component to an Entity-Architecture
- Widely Used in Simulation Environments
  - Provides a Flexible and Fast Path to Design Alternatives
- Limited or No Support in Synthesis Environments

```
CONFIGURATION <identifier> OF <entity_name> IS
        FOR <architecture_name>
        END FOR;
END; (1076-1987 version)
END CONFIGURATION; (1076-1993 version)
```


33

Putting It All Together

```
ENTITY cmpl_sig IS
PORT ( a, b, sel : IN bit;
       x, y, z : OUT bit);
END cmpl_sig;
ARCHITECTURE logic OF cmpl_sig IS
BEGIN
        -- simple signal assignment
      x <= (a AND NOT sel) OR (b AND sel);
        -- conditional signal assignment
      y <= a WHEN sel='0' ELSE
           b;
        -- selected signal assignment
      WITH sel SELECT
                z <= a WHEN '0',
                     b WHEN '1',
                    '0' WHEN OTHERS;
END logic;
CONFIGURATION cmpl_sig_conf OF cmpl_sig IS
      FOR logic
      END FOR;
END cmpl_sig_conf;
```



34

Packages

- Packages Are a Convenient Way of Storing and Using Information Throughout an Entire Model
- Packages Consist Of:
  - Package Declaration (Required)
    - Type Declarations
    - Subprograms Declarations
  - Package Body (Optional)
    - Subprogram Definitions
- VHDL Has Two Built-in Packages
  - Standard
  - Textio

35

Packages

PACKAGE <package_name> **IS**
Constant Declarations
Type Declarations
Signal Declarations
Subprogram Declarations
Component Declarations
--There are other Declarations
END <package_name> **;** (1076-1987)
END PACKAGE <package_name> **;** (1076-1993)
PACKAGE BODY <package_name> **IS**
Constant Declarations
Type Declarations
Subprogram Body
END <package_name> **;** (1076-1987)
END PACKAGE BODY <package_name> **;** (1076-1993)

36

Package Example

```
LIBRARY IEEE;
USE IEEE.std_logic_1164.all;
PACKAGE filt_cmp IS
   TYPE state_type IS (idle, tap1, tap2, tap3, tap4);
   COMPONENT acc
            port(xh : in std_logic_vector(10 downto 0);
                 clk, first: in std_logic;
                    yn : out std_logic_vector(11 downto 4));
   END COMPONENT;
FUNCTION compare (SIGNAL a , b : integer) RETURN boolean;
END filt_cmp;
PACKAGE BODY filt_cmp IS
FUNCTION compare (SIGNAL a , b : integer) RETURN boolean IS
   VARIABLE temp : boolean;
 Begin
            If a < b then
               temp := true;
            else
               temp := false;
             end if;
            RETURN temp;
END compare;
END filt_cmp;
```

Package Declaration

Package Body

37

Libraries

- Contains a Package or a Collection of Packages
- Resource Libraries
  - Standard Package
  - IEEE Developed Packages
  - Altera Component Packages
  - Any Library of Design Units That Are Referenced in a Design
- Working Library
  - Library Into Which the Unit Is Being Compiled

38

Model Referencing of Library/Package

- All Packages Must Be Compiled
- Implicit Libraries
  - Work
  - Std
  - ⇨ Note: Items in These Packages Do Not Need to Be Referenced, They Are Implied
- **LIBRARY** Clause
  - Defines the Library Name That Can Be Referenced
  - Is a Symbolic Name to Path/Directory
  - Defined by the Compiler Tool
- **USE** Clause
  - Specifies the Package and Object in the Library That You Have Specified in the Library Clause

39

Example

```
LIBRARY IEEE;
USE IEEE.std_logic_1164.all;
ENTITY cmpl_sig IS
PORT ( a, b, sel : IN std_logic;
       x, y, z : OUT std_logic);
END cmpl_sig;
ARCHITECTURE logic OF cmpl_sig IS
BEGIN
        -- simple signal assignment
      x <= (a AND NOT sel) OR (b AND sel);
        -- conditional signal assignment
      y <= a WHEN sel='0' ELSE
           b;
        -- selected signal assignment
      WITH sel SELECT
            z <= a WHEN '0',
                 b WHEN '1',
                 '0' WHEN OTHERS;
END logic;
CONFIGURATION cmpl_sig_conf OF cmpl_sig IS
      FOR logic
      END FOR;
END cmpl_sig_conf;
```

- **LIBRARY** <Name>, <Name> ;
  - Name Is Symbolic and Defined by Compiler Tool
  - ⇨ Note: Remember That WORK and STD Do Not Need to Be Defined.
- **Use** Lib_name.Pack_name.Object;
  - **All** Is a Reserved Word
- Placing the Library/Use Clause First Will Allow All Following Design Units to Access It

40

Libraries

- **Library** Std;
  - Contains the Following Packages:
    - **Standard** (Types: Bit, Boolean, Integer, Real, and Time; All Operator Functions to Support Types)
    - **Textio** (File Operations)
  - An Implicit Library (Built-in)
    - Does Not Need to Be Referenced in VHDL Design

41

Types Defined in Standard Package

- Type BIT
  - 2 Logic Value System ('0', '1')
    Signal A_temp : Bit;
  - Bit_vector Array of Bits
    Signal Temp : Bit_vector(3 **Downto** 0);
    Signal Temp : Bit_vector(0 **to** 3) ;
- Type Boolean
  - (False, True)
- Integer
  - Positive and Negative Values in Decimal
    Signal Int_tmp : Integer; -- 32 Bit Number
    Signal Int_tmp1 : Integer **Range** 0 to 255; --8 Bit Number

⇨ Note: Standard Package Has Other Types

42

Libraries

- **Library** IEEE;
  - Contains the Following Packages:
    - **Std_logic_1164** (Std_logic Types & Related Functions)
    - **Std_logic_arith** (Arithmetic Functions)
    - **Std_logic_signed** (Signed Arithmetic Functions)
    - **Std_logic_unsigned** (Unsigned Arithmetic Functions)

Types Defined in Std_logic_1164 Package

- Type **STD_LOGIC**
  - 9 Logic Value System ('U', 'X', '0', '1', 'Z', 'W', 'L', 'H', '-')
    - 'W', 'L', 'H" Weak Values (Not Supported by Synthesis)
    - 'X' - Used for Unknown
    - 'Z' - (Not 'z') Used for Tri-state
    - '-' Don't Care
  - Resolved Type: Supports Signals With Multiple Drives
- Type **STD_ULOGIC**
  - Same 9 Value System As STD_LOGIC
  - Unresolved Type: Does Not Support Multiple Signal Drives; Error Will Occur

User-defined Libraries/Packages

- User-defined Packages Can Be in the Same Directory As the Design

  Library Work; *--Optional*
  USE WORK.<*Package Name*>**.All**;

- Or Can Be in a Different Directory From the Design

  LIBRARY <*Any_name*>**;**
  Use <*Any_name*>.<*Package_name*>**.All;**

Architecture Modeling Fundamentals

Section Overview

- Understanding the Concept and Usage of Signals
  - Signal Assignments
  - Concurrent Signal Assignment Statements
  - Signal Delays
- Processes
  - Implied
  - Explicit
- Understanding the Concept and Usage of Variables
- Sequential Statement
  - If-then
  - Case
  - Loops

Using Signals

- Signals Represent Physical Interconnect (Wire) That Communicate Between Processes (Functions)
- Signals Can Be Declared in **Packages**, **Entity** and **Architecture**



Assigning Values to Signals

SIGNAL temp : **STD_LOGIC_VECTOR (**7 **downto** 0**);**

- All Bits:
  Temp **<=** "10101010"**;**
  Temp **<= X**"aa" **;** (1076-1993)
- Single Bit:
  Temp(7) **<=** '1';
- Bit-slicing:
  Temp (7 Downto 4) **<=** "1010";
- Single-bit: Single-quote (')
- Multi-bit: Double-quote (")

Signal Used As an Interconnect

```
LIBRARY IEEE;
USE IEEE.std_logic_1164.all;
ENTITY simp IS
PORT(r, t, g, h : IN STD_LOGIC;
        qb : OUT STD_LOGIC);
END simp;
ARCHITECTURE logic OF simp IS
SIGNAL qa : STD_LOGIC;

BEGIN

qa <= r or t;
qb <= (qa and not(g xor h));

END logic;
```



- **r, t, g, h**, and **qb** Are Signals (by Default)
- **qa** Is a Buried Signal and Needs to Be Declared

Signal Declaration Inside Architecture

Signal Assignments

- Signal Assignments Are Represented By: **<=**
- Signal Assignments Have an ***Implied*** Process (Function) That Synthesizes to Hardware



Signal Assignment **<=** Implied Process

Concurrent Signal Assignments

- Three Concurrent Signal Assignments:
  - Simple Signal Assignment
  - Conditional Signal Assignment
  - Selected Signal Assignment

52

Simple Signal Assignments

- Format: *<signal_name>* **<=** *<expression>***;**
- Example:

```
qa <= r or t ;
qb <= (qa and not(g xor h));
```

Implied Process

⇨ Parenthesis **( )** *Give the Order of Operation*



- VHDL Operators Are Used to Describe the Process

53

VHDL Operators

Operator Type	Operator Name/Symbol
Logical	**and or nand nor xor xnor**(1)
Relational	**= /= < <= > >=**
Addition & Concatenation	**+ - &**
Signing	**+ -**
Multiplying	*** / mod rem**
Miscellaneous	**** abs not**

(1) Supported in VHDL '93 Only

54

VHDL Operators

- VHDL Defines Arithmetic & Boolean Functions Only for Built-in Data Types (Defined in ***Standard*** Package)
  - Arithmetic Operators Such As **+**, **-**, **<**, **>**, **<=**, **>=** Are Defined ***Only*** for **INTEGER** Type
  - Boolean Operators Such As **AND**, **OR**, **NOT** Are Defined ***Only*** for **BIT** Type
- Recall: VHDL Implicit Library (Built-in)
  - **Library STD**
    - Types Defined in the **Standard** Package:
      - **Bit, Boolean, Integer**
  - ⇨ Note: Items in This Package Do Not Need to Be Referenced, They Are Implied

55

Arithmetic Function

```
ENTITY opr IS
        PORT ( a   : IN  INTEGER RANGE 0 TO 16;
               b   : IN  INTEGER RANGE 0 TO 16;
               sum : OUT  INTEGER RANGE 0 TO 32);
END opr;

ARCHITECTURE example OF opr IS
BEGIN
        sum <= a + b;
END example;
```

The VHDL Compiler Can Understand This Operation Because an Arithmetic Operation Is Defined for the Built-in Data Type ***Integer***

⇨ Note: Remember the Library **STD** and the Package **Standard** Do Not Need to Be Referenced

56

Operator Overloading

- How Do You Use Arithmetic & Boolean Functions With Other Data Types?
  - ***Operator Overloading*** - Defining Arithmetic & Boolean Functions With Other Data Types
- Operators Are Overloaded by Defining a Function Whose Name Is the Same As the Operator Itself
  - Because the Operator and Function Name Are the Same, the Function Name Must Be Enclosed Within Double Quotes to Distinguish It From the Actual VHDL Operator
  - The Function Is Normally Declared in a Package So That It Is Globally Visible for Any Design

57

Operator Overloading Function/package

- Packages That Define These Operator Overloading Functions Can Be Found in the **LIBRARY IEEE**
- For Example, the Package ***std_logic_unsigned*** Defines Some of the Following Functions

```
package std_logic_unsigned is

  function "+"(L: STD_LOGIC_VECTOR; R: STD_LOGIC_VECTOR) return STD_LOGIC_VECTOR;
  function "+"(L: STD_LOGIC_VECTOR; R: INTEGER) return STD_LOGIC_VECTOR;
  function "+"(L: INTEGER; R: STD_LOGIC_VECTOR) return STD_LOGIC_VECTOR;
  function "+"(L: STD_LOGIC_VECTOR; R: STD_LOGIC) return STD_LOGIC_VECTOR;
  function "+"(L: STD_LOGIC; R: STD_LOGIC_VECTOR) return STD_LOGIC_VECTOR;

  function "-"(L: STD_LOGIC_VECTOR; R: STD_LOGIC_VECTOR) return STD_LOGIC_VECTOR;
  function "-"(L: STD_LOGIC_VECTOR; R: INTEGER) return STD_LOGIC_VECTOR;
  function "-"(L: INTEGER; R: STD_LOGIC_VECTOR) return STD_LOGIC_VECTOR;
  function "-"(L: STD_LOGIC_VECTOR; R: STD_LOGIC) return STD_LOGIC_VECTOR;
  function "-"(L: STD_LOGIC; R: STD_LOGIC_VECTOR) return STD_LOGIC_VECTOR;
```


58

Use of Operator Overloading

```
LIBRARY IEEE;
USE IEEE.std_logic_1164.all;
USE IEEE.std_logic_unsigned.all;
```

Include These Statements At the Beginning of a Design File

```
ENTITY overload IS
        PORT ( a   : IN STD_LOGIC_VECTOR (4 downto 0);
               b   : IN STD_LOGIC_VECTOR (4 downto 0);
               sum : OUT STD_LOGIC_VECTOR (4 downto 0));
END overload;

ARCHITECTURE example OF overload IS
BEGIN
        sum <= a + b;
END example;
```

This Allows Us to Perform Arithmetic on Non-built-in Data Types

59

Exercise 1

Please go to Exercise 1

60

Concurrent Signal Assignments

- Three Concurrent Signal Assignments:
  - Simple Signal Assignment
  - Conditional Signal Assignment
  - Selected Signal Assignment

61

Conditional Signal Assignments

- Format:

```
<signal_name> <= <signal/value> when <condition1> else
                 <signal/value> when <condition2> else
                                .
                                .
                 <signal/value> when <condition3> else
                 <signal/value>;
```

- Example:

```
q <= a WHEN sela = '1' ELSE
     b WHEN selb = '1' ELSE
     c;
```



Implied Process

62

Selected Signal Assignments

- Format:

```
with <expression> select
<signal_name> <=    <signal/value> when <condition1>,
                    <signal/value> when <condition2>,
                                   .
                                   .
                    <signal/value> when others;
```

- Example:

```
WITH sel SELECT
q <= a WHEN "00",
     b WHEN "01",
     c WHEN "10",
     d WHEN OTHERS;
```



Implied Process

63

Selected Signal Assignments

- **All** Possible Conditions Must Be Considered
- **WHEN OTHERS** Clause Evaluates All Other Possible Conditions That Are Not Specifically Stated

SEE NEXT SLIDE

64

Selected Signal Assignment

```
LIBRARY IEEE;
USE IEEE.std_logic_1164.all;

ENTITY cmpl_sig IS
PORT ( a, b, sel : IN STD_LOGIC;
       z : OUT STD_LOGIC);
END cmpl_sig;

ARCHITECTURE logic OF cmpl_sig IS
BEGIN
    -- selected signal assignment
    WITH sel SELECT
        z <= a WHEN '0',
             b WHEN '1',
             '0' WHEN OTHERS;
END logic;
```

sel Has a ***STD_LOGIC*** Data Type

- *What are the Values for a **STD_LOGIC** Data Type*
- *Answer:* **{'0','1','X','Z'}**
- *Therefore, is the **WHEN OTHERS** Clause Necessary?*
- *Answer:* **YES**

65

VHDL Model - Concurrent Signal Assignments

```
LIBRARY IEEE;
USE IEEE.std_logic_1164.all;

ENTITY cmpl_sig IS
PORT ( a, b, sel : IN STD_LOGIC;
       x, y, z : OUT STD_LOGIC);
END cmpl_sig;

ARCHITECTURE logic OF cmpl_sig IS
BEGIN
    -- simple signal assignment
    x <= (a AND NOT sel) OR (b AND sel);
    -- conditional signal assignment
    y <= a WHEN sel='0' ELSE
         b;
    -- selected signal assignment
    WITH sel SELECT
        z <= a WHEN '0',
             b WHEN '1',
             '0' WHEN OTHERS;
END logic;
```

- The Signal Assignments Execute in Parallel, and Therefore the Order We List the Statements Should Not Affect the Outcome



66

Explicit Process Statement

- Process Can Be Thought of As
  - *Implied Processes*
  - *Explicit Processes*
- Implied Process Consist of
  - Concurrent Signal Assignment Statements
  - Component Statements
  - Processes' Sensitivity Is Read Side of Expression
- Explicit Process
  - Concurrent Statement
  - Consist of Sequential Statements Only

```
--  Explicit Process Statement
PROCESS (sensitivity_list)
   Constant Declarations
   Type Declarations
   Variable Declarations
        BEGIN
        -- Sequential statement #1;
        -- ……..
        -- Sequential statement #N ;
        END PROCESS;
```


67

Execution of Process Statement

- Process Statement Is Executed Infinitely Unless Broken by a WAIT Statement or Sensitivity List
  - Sensitivity List Implies a WAIT Statement at the End of the Process
  - Process Can Have Multiple WAIT Statements
  - Process Can Not Have Both a Sensitivity List and WAIT Statement

  ⇨ Note: Logic Synthesis Places Restrictions on WAIT and Sensitivity List

```
PROCESS (a,b)
  BEGIN
    --sequential statements
  END PROCESS;
```

```
PROCESS
    BEGIN
       -- sequential statements
    WAIT ON (a,b) ;
     END PROCESS;
```


68

Multi-Process Statements



- An Architecture Can Have Multiple Process Statements
- Each Process Executes in Parallel With Each Other
- However, Within a Process, the Statements Are Executed Sequentially

69

VHDL Model - Multi-Process Architecture

```
LIBRARY IEEE;
USE IEEE.std_logic_1164.all;

ENTITY if_case IS
PORT ( a, b, c, d : IN STD_LOGIC;
       sel : IN STD_LOGIC_VECTOR(1 DOWNTO 0);
       y, z : OUT STD_LOGIC);
END if_case;

ARCHITECTURE logic OF if_case IS
BEGIN
if_label: PROCESS(a, b, c, d, sel)
        BEGIN
                IF sel="00" THEN
                        y <= a;
                ELSIF sel="01" THEN
                        y <= b;
                ELSIF sel="10" THEN
                        y <= c;
                ELSE
                        y <= d;
                END IF;
END PROCESS if_label;

case_label: PROCESS(a, b, c, d, sel)
                BEGIN
                  CASE sel IS
                        WHEN "00" =>
                                z <= a;
                        WHEN "01" =>
                                z <= b;
                        WHEN "10" =>
                                z <= c;
                        WHEN "11" =>
                                z <= d;
                        WHEN OTHERS =>
                                z <= '0';
                  END CASE;
END PROCESS case_label;
END logic;
```

- The Process Statements Execute in Parallel and Therefore, the Order in Which We List the Statements Should Have No Affect on the Outcome
- Within a Process, the Statements Are Executed Sequentially
- Signal Assignments Can Also Be Inside Process Statements

Signal Assignment - Delay

- Signal Assignments Can Be Inside Process Statements or Outside (Like the Three Concurrent Signal Assignments)
- Signal Assignments Incur Delay
  - Two Types of Delays
    - Inertial Delay (Default)
      - A Pulse That Is Short in Duration of the Propagation Delay Will Not Be Transmitted
      - Ex. `a <= b AFTER 10 ns;`
    - Transport Delay
      - Any Pulse Is Transmitted No Matter How Short
      - Ex. `a <= TRANSPORT b AFTER 10 ns;`
  - ⇨ In VHDL, There Are Exceptions to This Rule That Will Not Be Discussed

VHDL Simulation

- Event - A Change in Value: From 0 to 1; Or From X to 1, Etc.
- Simulation Cycle
  - Wall Clock Time
  - Delta
    - Process Execution Phase
    - Signal Update Phase
- When Does a Simulation Cycle End and a New One Begin?
  - ⇨ **When:**
    - **All Processes Execute**
    - **Signals Are Updated**
- Signals Get Updated at the End of the Process



Equivalent Functions

```
LIBRARY IEEE;
USE IEEE.std_logic_1164.all;
ENTITY simp IS
PORT(a, b : IN STD_LOGIC;
          y : OUT STD_LOGIC);
END simp;
ARCHITECTURE logic OF simp IS
SIGNAL c : STD_LOGIC;

BEGIN

c <= a and b;
y <= c;

END logic;
```

```
LIBRARY IEEE;
USE IEEE.std_logic_1164.all;
ENTITY simp_prc IS
PORT(a,b : IN STD_LOGIC;
          y : OUT STD_LOGIC);
END simp_prc;
ARCHITECTURE logic OF simp_prc IS
SIGNAL c : STD_LOGIC;

BEGIN
process1: PROCESS(a, b)
          BEGIN
                    c <= a and b;
          END PROCESS process1;
process2: PROCESS(c)
          BEGIN
                    y <= c;
          END PROCESS process2;
END logic;
```

- **c** and **y** Get Executed and Updated in Parallel at the End of the Process Within One Simulation Cycle

73

Equivalent Functions?

```
LIBRARY IEEE;
USE IEEE.std_logic_1164.all;
ENTITY simp IS
PORT(a, b : IN STD_LOGIC;
          y : OUT STD_LOGIC);
END simp;
ARCHITECTURE logic OF simp IS
SIGNAL c : STD_LOGIC;
BEGIN

c <= a and b;

y <= c;

END logic;
```

```
LIBRARY IEEE;
USE IEEE.std_logic_1164.all;
ENTITY simp_prc IS
PORT(a, b : IN STD_LOGIC;
          y: OUT STD_LOGIC);
END simp_prc;
ARCHITECTURE logic OF simp_prc IS
SIGNAL c: STD_LOGIC;

BEGIN
PROCESS(a, b)
          BEGIN
          c <= a and b;
          y <= c;
END PROCESS;
END logic;
```


74

Variable Declarations

- Variables Are Declared Inside a Process
- Variables Are Represented By: **:=**
- Variable Declaration
  VARIABLE <*Name*> **:** <*DATA_TYPE*> **:=** <*Value*>**;**
  Variable Temp **: Std_logic_vector** (7 **Downto** 0)**;**
- Variable Assignments Are Updated Immediately
  - Do Not Incur a Delay



75

Assigning Values to Variables

VARIABLE temp : **STD_LOGIC_VECTOR** (7 **downto** 0**);**

- All Bits:
  - Temp **:=** "10101010"**;**
  - Temp **:= X**"aa" ; (1076-1993)
- Single Bit:
  - Temp(7) **:=** '1';
- Bit-slicing:
  - Temp (7 downto 4) **:=** "1010";
- Single-bit: Single-quote (')
- Multi-bit: Double-quote (")

Variable Assignment

```
LIBRARY IEEE;
USE IEEE.std_logic_1164.all;

ENTITY var IS
PORT    (a, b : IN  STD_LOGIC;
         y : OUT  STD_LOGIC);
END var;

ARCHITECTURE logic OF var IS
BEGIN

        PROCESS (a, b)
        VARIABLE  c  :  STD_LOGIC;
        BEGIN
        c := a AND b;

        y <= c;

        END PROCESS;
END logic;
```

Variable Declaration

Variable Assignment

***Variable** is Assigned to a **Signal** to Synthesize to a Piece of Hardware*

Use of a Variable

```
LIBRARY IEEE;
USE IEEE.std_logic_1164.all;
ENTITY cmb_var IS
PORT(i0, i1, a : IN BIT;
        q : OUT BIT);
END cmb_var;
ARCHITECTURE logic OF cmb_var IS
BEGIN
        PROCESS(i0, i1, a)
        VARIABLE val : INTEGER RANGE 0 TO 1;
        BEGIN
                val := 0;
                IF (a = '0') THEN
                        val := val;
                ELSE
                        val := val + 1;
                END IF;
                CASE val IS
                        WHEN 0 =>
                                q <= i0;

                        WHEN 1 =>
                                q <= i1;
                END CASE;
        END PROCESS;
END logic;
```

***val** Is a Variable That Is Updated at the Instant an Assignment Is Made to It*

*Therefore, the Updated Value of **val** Is Available for the CASE Statement*

Signal and Variable Scope



79

Review - Signals vs. Variables

	SIGNALS (<=)	VARIABLES (:=)
ASSIGN	assignee <= assignment	assignee := assignment
UTILITY	Represent Circuit Interconnect	Represent Local Storage
SCOPE	Global Scope (Communicate Between PROCESSES)	Local Scope (Inside PROCESS)
BEHAVIOR	Updated at End of Process Statement (New Value Not Available)	Updated Immediately (New Value Available)

80

Sequential Statements

- Sequential Statements
  - IF-THEN Statement
  - CASE Statement
  - Looping Statements

81

If-then Statements

- Format:

```
IF <condition1> THEN
        {sequence of statement(s)}
ELSIF <condition2> THEN
        {sequence of statement(s)}
                .
                .
ELSE
        {sequence of statement(s)}
END IF;
```

- Example:

```
PROCESS(sela, selb, a, b, c)
BEGIN
   IF sela='1' THEN
        q <= a;
   ELSIF selb='1' THEN
        q <= b;
   ELSE
        q <= c;
   END IF;
END PROCESS;
```



82

If-then Statements

- Conditions Are Evaluated in Order From Top to Bottom
  - Prioritization
- The First Condition That Is True Causes the Corresponding Sequence of Statements to Be Executed
- If All Conditions Are False, Then the Sequence of Statements Associated With the "ELSE" Clause Is Evaluated

83

If-then Statements

- Similar to Conditional Signal Assignment

Implied Process

```
q <= a WHEN sela = '1' ELSE
     b WHEN selb = '1' ELSE
     c;
```



Explicit Process

```
PROCESS(sela, selb, a, b, c)
BEGIN
   IF sela='1' THEN
        q <= a;
   ELSIF selb='1' THEN
        q <= b;
   ELSE
        q <= c;
   END IF;
END PROCESS;
```


84

Exercise 2

Please go to Exercise 2

Case Statement

- Format:

```
CASE {expression} IS
        WHEN <condition1> =>
                {sequence of statements}
        WHEN <condition2> =>
                {sequence of statements}
                .
                .
        WHEN OTHERS =>  -- (optional)
                {sequence of statements}
        END CASE;
```

- Example:

```
PROCESS(sel, a, b, c, d)
BEGIN
    CASE sel IS
        WHEN "00" =>
                q <= a;
        WHEN "01" =>
                q <= b;
        WHEN "10" =>
                q <= c;
        WHEN OTHERS =>
                q <= d;
    END CASE;
END PROCESS;
```



Case Statement

- Conditions Are Evaluated at Once
  - No Prioritization
- **All** Possible Conditions Must Be Considered
- **WHEN OTHERS** Clause Evaluates All Other Possible Conditions That Are Not Specifically Stated

Case Statement

- Similar to Selected Signal Assignment

Implied Process

```
WITH sel SELECT
        q <= a WHEN "00",
             b WHEN "01",
             c WHEN "10",
             d WHEN OTHERS;
```



Explicit Process

```
PROCESS(sel, a, b, c, d)
BEGIN
   CASE sel IS
        WHEN "00" =>
             q <= a;
        WHEN "01" =>
             q <= b;
        WHEN "10" =>
             q <= c;
        WHEN OTHERS =>
             q <= d;
   END CASE;
END PROCESS;
```


88

Exercise 3

Please go to Exercise 3

89

Sequential LOOPS

- Infinite Loop
  - Loops Infinitely Unless EXIT Statement Exists

```
[loop_label]LOOP
  --sequential statement
 EXIT loop_label ;
END LOOP;
```

- While Loop
  - Conditional Test to End Loop

```
WHILE <condition> LOOP
  --sequential statements
END LOOP;
```

- For Loop
  - Iteration Loop

```
FOR <identifier> IN <range> LOOP
  --sequential statements
END LOOP;
```


90

FOR LOOP Using a Variable: 4-bit Left Shifter

```
LIBRARY IEEE;
USE IEEE.std_logic_1164.all;
USE IEEE.std_logic_unsigned.all;
ENTITY shift4 IS
PORT (  shft_lft : in std_logic;
                d_in : in std_logic_vector(3 downto 0);
                q_out : out std_logic_vector(7 downto 0));
END shift4;
ARCHITECTURE logic OF shift4 IS
BEGIN

PROCESS(d_in, shft_lft)
        VARIABLE shft_var : std_logic_vector(7 DOWNTO 0);
BEGIN
        shft_var(7 downto 4) := "0000";
        shft_var(3 downto 0) := d_in;
```

Variable Declaration

Variable is Initialized

91

FOR LOOP Using a Variable: 4-bit Left Shifter

```
IF shft_lft = '1' THEN
        FOR i IN 7 DOWNTO 4 LOOP
                shft_var(i) := shft_var(i-4);
        END LOOP;
                shft_var(3 downto 0) := "0000";
ELSE
        shft_var := shft_var;
END IF;

q_out <= shft_var;

END PROCESS;
END logic;
```

Enables Shift-Left

i Is the Index for the FOR LOOP and Does Not Need to Be Declared

Shifts Left by 4

Fills the LSBs with Zeros

No Shifting

Variable is Assigned to a **Signal** Before the End of the Process to Synthesize to a Piece of Hardware

92

FOR LOOP: '1's Counter

```
LIBRARY IEEE;
USE IEEE.std_logic_1164.all;
USE IEEE.std_logic_unsigned.all;
USE IEEE.std_logic_arith.all;

ENTITY bc IS PORT (invec: in std_logic_vector(31 downto 0);
        outvec: out std_logic_vector(7 downto 0));
END bc;
ARCHITECTURE rtl OF bc IS
BEGIN
PROCESS(invec)
        VARIABLE count: std_logic_vector(7 downto 0);
```

Variable Declaration

93

FOR LOOP: '1's Counter

```
BEGIN
    Count:=(others=>'0');
    FOR i IN invec'right TO Invec'left LOOP
        IF (invec(i)='0') THEN
            count:=count+1;
        END IF;
    END LOOP;
outvec<=count;
END PROCESS;
END rtl;
```

Variable Is Initialized

i Is the Loop Index
This Loop Examines All 32 Bits of invec. If the Current Bit Does Not Equal Zero, Count is Incremented.

Variable is Assigned to a **Signal** Before the End of the Process to Synthesize to a Piece of Hardware

94

Exercise 4

Please go to Exercise 4

95

Understanding VHDL and Logic Synthesis

96

VHDL Model - RTL Modeling

Result:



- **RTL** - Type of Behavioral Modeling that Implies or Infers Hardware
- Functionality and Somewhat Structure of the Circuit
- For the Purpose of Synthesis, As Well As Simulation

97

Recall - RTL Synthesis

```
IF sel="00" THEN
    mux_out <= a;
ELSIF sel="01" THEN
    mux_out <= b;
.....
ELSE sel="11" THEN
    mux_out <= d;
```

Inferred



Translation



Optimization



98

Two Types of Process Statements

- **Combinatorial Process**
  - Sensitive to All Inputs Used In the Combinatorial Logic
- **Example**

```
PROCESS(a, b, sel)
```

Sensitivity List Includes All Inputs Used in the Combinatorial Logic



- **Sequential Process**
  - Sensitive to a Clock or/and Control Signals
- **Example**

```
PROCESS(clr, clk)
```

Sensitivity List Does Not Include the **d** *Input, Only the Clock or/and Control Signals*



99

Latch

```
LIBRARY IEEE;
USE IEEE.std_logic_1164.ALL;

ENTITY latch1 IS
PORT ( data : IN std_logic;
                gate : IN std_logic;
                q : OUT std_logic
        );
END latch1;

ARCHITECTURE behavior OF latch1 IS
BEGIN
label_1: PROCESS (data, gate)
        BEGIN
        IF gate = '1' THEN
                q <= data;
        END IF;
END PROCESS;

END behavior;
```



Sensitivity List Includes Both Inputs

What Happens if Gate = '0'?
⇨ ***Implicit Memory***

100

DFF With WAIT Statement

```
LIBRARY IEEE;
USE IEEE.std_logic_1164.all;

ENTITY wait_dff IS
PORT ( d, clk : in std_logic;
                q : out std_logic
        );
END wait_dff;

ARCHITECTURE behavior OF wait_dff IS
BEGIN
PROCESS
        BEGIN
        wait until clk = '1';
                q <= d;
END PROCESS;
END behavior;
```



Note: There is No Sensitivity List

wait until
- *Acts Like the Sensitivity List*

101

DFF - Clk'event and Clk='1'

```
LIBRARY IEEE;
USE IEEE.std_logic_1164.all;

ENTITY dff_a IS
PORT ( d : in std_logic;
                clk : in std_logic;
                q : out std_logic
        );
END dff_a;

ARCHITECTURE behavior OF dff_a IS
BEGIN
PROCESS (clk)
        BEGIN
        IF clk'event and clk = '1' THEN
                q <= d;
        END IF;
END PROCESS;
END behavior;
```



clk'event* and *clk='1'
- ***clk*** *Is the Signal Name (Any Name)*
- ***'event*** *Is a VHDL Attribute, Specifying That There Needs to Be a Change in Signal Value*
- ***clk='1'*** *Means Positive-Edge Triggered*

102

DFF - Rising_edge

```
LIBRARY IEEE;
USE IEEE.std_logic_1164.all;

ENTITY dff_b IS
PORT ( d : in std_logic;
                clk : in std_logic;
                q : out std_logic
        );
END dff_b;

ARCHITECTURE behavior OF dff_b IS
BEGIN
PROCESS(clk)
        BEGIN
        IF rising_edge(clk) THEN
                q <= d;
        END IF;
END PROCESS;
END behavior;
```



rising_edge
- *IEEE Function That is Defined in the std_logic_1164 Package*
- *Specifies That the Signal Value **must** be 0 to 1*
- *X, Z to 1 Transition Is Not Allowed*

103

DFF With Asynchronous Clear

```
LIBRARY IEEE;
USE IEEE.std_logic_1164.all;
USE IEEE.std_logic_unsigned.all;

ENTITY dff_clr IS
PORT (  clr : in bit;
        d, clk : in std_logic;
        q : out std_logic
        );
END dff_clr;

ARCHITECTURE behavior OF dff_clr IS
BEGIN
PROCESS(clk, clr)
        BEGIN
        IF clr = '0' THEN
                q <= '0';
        ELSIF rising_edge(clk) THEN
                q <= d;
        END IF;
END PROCESS;
END behavior;
```



- *This is How to Implement Asynchronous Control Signals for the Register*
- *Note: This IF-THEN Statement Is Outside the IF-THEN Statement that Checks the Condition **rising_edge***
- *Therefore, **clr='1'** Does Not Depend on the Clock*

104

How Many Registers?

```
LIBRARY IEEE;
USE IEEE.std_logic_1164.all;
USE IEEE.std_logic_unsigned.all;
ENTITY reg1 IS
        PORT ( d        : in STD_LOGIC;
                clk     : in STD_LOGIC;
                q       : out STD_LOGIC);
END reg1;

ARCHITECTURE reg1 OF reg1 IS
SIGNAL a, b : STD_LOGIC;
BEGIN
        PROCESS (clk)
        BEGIN
                IF rising_edge(clk) THEN
                        a <= d;
                        b <= a;
                        q <= b;
                END IF;
        END PROCESS;
END reg1;
```


105

How Many Registers?

- Signal Assignments Inside the IF-THEN Statement That Checks the Clock Condition Infer Registers



106

How Many Registers?

```
LIBRARY IEEE;
USE IEEE.std_logic_1164.all;
USE IEEE.std_logic_unsigned.all;
ENTITY reg1 IS
      PORT ( d     : in STD_LOGIC;
             clk   : in STD_LOGIC;
             q     : out STD_LOGIC);
END reg1;
ARCHITECTURE reg1 OF reg1 IS
SIGNAL a, b : STD_LOGIC;
BEGIN
      PROCESS (clk)
      BEGIN
            IF rising_edge(clk)  THEN
                  a <= d;
                  b <= a;
            END IF;
      END PROCESS;
      q <= b;
END reg1;
```

Signal Assignment Moved

107

How Many Registers?

- B to Q Assignment Is No Longer Edge-sensitive Because It Is Not Inside the If-then Statement That Checks the Clock Condition



108

How Many Registers?

```
LIBRARY IEEE;
USE IEEE.std_logic_1164.all;
USE IEEE.std_logic_unsigned.all;
ENTITY reg1 IS
        PORT ( d        : in STD_LOGIC;
                clk     : in STD_LOGIC;
                q       : out STD_LOGIC);
END reg1;

ARCHITECTURE reg1 OF reg1 IS
BEGIN
        PROCESS (clk)
        VARIABLE a, b : STD_LOGIC;
        BEGIN
                IF rising_edge(clk) THEN
                        a := d;
                        b := a;
                        q <= b;
                END IF;
        END PROCESS;
END reg1;
```

Signals Changed to Variables

109

How Many Registers?

- Variable Assignments Are Updated Immediately
- Signal Assignments Are Updated on Clock Edge



110

Variable Assignments in Sequential Logic

- Variable Assignments Inside the IF-THEN Statement, That Checks the Clock Condition, Usually Don't Infer Registers
  - Exception: If the Variable Is on the Right Side of the Equation in a Clocked Process Prior to Being Assigned a Value, the Variable Will Infer a Register(s)
- Variable Assignments Are Temporary Storage and Have No Hardware Intent
- Variable Assignments Can Be Used in Expressions to Immediately Update a Value
  - Then the Variable Can Be Assigned to a Signal

111

Example - Counter Using a Variable

- Counters Are Accumulators That Always Add a '1' or Subtract a '1'
- This Example Takes 17 LEs

```
LIBRARY IEEE;
USE IEEE.std_logic_1164.all;
USE IEEE.std_logic_unsigned.all;
ENTITY count_a IS
PORT (clk, rst, updn : in std_logic;
        q : out std_logic_vector(15 downto 0));
END count_a;
ARCHITECTURE logic OF count_a IS
BEGIN
PROCESS(rst, clk)
VARIABLE tmp_q : std_logic_vector(15 downto 0);
BEGIN
        IF rst = '0' THEN
                tmp_q := (others => '0');
        ELSIF rising_edge(clk) THEN
                IF updn = '1' THEN
                        tmp_q := tmp_q + 1;
                ELSE
                        tmp_q := tmp_q - 1;
                END IF;
        END IF;
        q <= tmp_q;
END PROCESS;
END logic;
```

Arithmetic Expression Assigned to a Variable

Variable Assigned to a Signal

112

Exercise 5

Please go to Exercise 5

113

Model Application

114

Finite State Machine (Fsm) - State Diagram



115

Enumerated Data Type

- Recall the Built-in Data Types:
  - Bit
  - Std_logic
  - Integer
- What About User-defined Data Types?:
  - Enumerated Data Type:

TYPE <*your_data_type*> **IS**
(*items or values for your data type separated by commas*)

116

Writing VHDL Code for FSM

- State Machine States Must Be an Enumerated Data Type:
  TYPE *State_type* **IS** (Idle, Tap1, Tap2, Tap3, Tap4);
- Object Which Stores the Value of the Current State Must Be a ***Signal*** of the User-defined Type:
  SIGNAL Filter : *State_type*;

117

Writing VHDL Code for FSM

- To Determine Next State Transition/Logic:
  - Use a **CASE** Statement Inside IF-THEN Statement That Checks for the Clock Condition
    - Remember: State Machines Are Implemented Using Registers
- To Determine State Machine Outputs:
  - Use **Conditional** and/or **Selected** Signal Assignments
  - Or Use a Second **Case** Statement to Determine the State Machine Outputs

118

FSM VHDL Code - Enumerated Data Type

```
LIBRARY IEEE;
USE IEEE.std_logic_1164.all;
USE IEEE.std_logic_unsigned.all;
USE IEEE.std_logic_arith.all;

ENTITY state_m2 IS
PORT(clk, reset, nw : in std_logic;
          sel: out std_logic_vector(1 downto 0);
          nxt, first: out std_logic);
END state_m2;

ARCHITECTURE logic OF state_m2 IS
          TYPE state_type IS
                    (idle, tap1, tap2, tap3, tap4);
          SIGNAL filter : state_type;
```

Enumerated Data Type



119

FSM VHDL Code - Next State Logic

```
BEGIN
PROCESS (reset, clk)
          BEGIN
          IF reset = '1' THEN
                    filter <= idle;
          ELSIF clk'event and clk = '1' THEN
                    CASE filter IS
                    WHEN idle =>
                              IF nw = '1' THEN
                                        filter <= tap1;
                              END IF;
                    WHEN tap1 =>
                              filter <= tap2;
                    WHEN tap2 =>
                              filter <= tap3;
                    WHEN tap3 =>
                              filter <= tap4;
                    WHEN tap4 =>
                              IF nw = '1' THEN
                                        filter <= tap1;
                              ELSE
                                        filter <= idle;
                              END IF;

                    END CASE;
          END IF;
END PROCESS;
```



120

FSM VHDL Code - Outputs

```
nxt <= '1' WHEN filter=tap4 ELSE
          '0';

first <= '1'  WHEN filter=tap1 ELSE
             '0';

WITH filter SELECT
        sel <= "00" WHEN tap1,
               "01" WHEN tap2,
               "10" WHEN tap3,
               "11" WHEN tap4,
               "00" WHEN others;

END logic;
```



Conditional Signal Assignments

Selected Signal Assignments

121

FSM VHDL Code - Outputs Using a Case

```
output: PROCESS(filter)
        BEGIN
        CASE  filter IS
                WHEN idle =>
                        nxt <= '0';
                        first <= '0';
                WHEN tap1 =>
                        sel <= "00";
                        first <= '1';
                WHEN tap2 =>
                        sel <= "01";
                        first <= '0';
                WHEN tap3 =>
                        sel <= "10";
                WHEN tap4 =>
                        sel <= "11";
                        nxt <= '1';
                END CASE;
END PROCESS output;

END logic;
```



122

Designing Hierarchically

123

Recall - Structural Modeling

- Functionality and Structure of the Circuit
- Call Out the Specific Hardware, Lower-Level Components
- For the Purpose of Synthesis



Design Hierarchically - Multiple Design Files

- VHDL Hierarchical Design Requires Component Declarations and Component Instantiations



Component Declaration and Instantiation

- Component Declaration - Used to Declare the *Port Type*s and the *Data Types* of the Ports for a Lower-level Design

```
COMPONENT <Lower-level_design_name> IS
PORT ( <Port_name> : <Port_type> <Data_type>;
                .
                .
       <Port_name> : <Port_type> <Data_type>);
END COMPONENT;
```

- Component Instantiation - Used to Map the Ports of a Lower-level Design to That of the Current-level Design

```
<Instance_name> : <Lower-level_design_name>
PORT MAP(<lower-level_port_name> => <Current_level_port_name>,
         ...,<Lower-level_port_name> =>
<Current_level_port_name>);
```

Component Declaration and Instantiation

- Upper-level of Hierarchy Design Must Have a Component Declaration for a Lower-level Design Before It Can Be Instantiated

```
ARCHITECTURE tolleab_arch OF tolleab IS
COMPONENT tollv
PORT(   clk : IN   STD_LOGIC;
        cross, nickel, dime, quarter  : IN  STD_LOGIC;
        green, red  : OUT  STD_LOGIC;
        sout  : OUT STATE_TYPE;
        state_in    : IN  STATE_TYPE);
END COMPONENT;
BEGIN
u1 :  tollv  PORT MAP ( tclk, tcross, tnickel, tdime,
                        tquarter, tgreen, tred,
                        tsout, tstate);
```

Component Declaration

Positional Association

Instance Label/Name

Component Instantiation

127

Component Declaration and Instantiation

```
LIBRARY IEEE;
USE  IEEE.std_logic_1164.all;
ENTITY tolleab IS
PORT(         tclk : IN STD_LOGIC;
                        tcross, tnickel, tdime, tquarter : IN STD_LOGIC;
                        tgreen, tred : OUT STD_LOGIC);
END tolleab;
ARCHITECTURE tolleab_arch OF tolleab IS
TYPE STATE_TYPE IS (cent0, cent5, cent10, cent15, cent20, cent25, cent30,
                    cent35, cent40, cent45, cent50, arrest);
SIGNAL connect : STATE_TYPE;

COMPONENT tollv
PORT(         clk: IN STD_LOGIC;
              cross, nickel, dime, quarter : IN STD_LOGIC;
              green, red : OUT STD_LOGIC;
              sout : OUT STATE_TYPE;
              state_in : IN STATE_TYPE);
END COMPONENT;

BEGIN

u1 :  tollv PORT MAP (clk => tclk, cross => tcross, nickel => tnickel, dime => tdime,
                      quarter => tquarter, green => tgreen, red => tred,
                      sout => connect, state_in => connect);

END tolleab_arch;
```

Lower-Level Port

dime => tdime

Current-Level Port

128

Benefits of Hierarchical Designing

Designing Hierarchically

- In a Design Group, Each Designer Can Create Separate Functions (Components) in Separate Design Files
- These Components Can Be Shared by Other Designers or Can Be Used for Future Projects
- Therefore, Designing Hierarchically Can Make Designs More Modular and Portable
- Designing Hierarchically Can Also Allow Easier and Faster Alternative Implementations
  - Example: Try Different Counter Implementations by Replacing Component Declaration and Component Instantiation

129

Vendor Libraries

- Silicon Vendors Often Provide Libraries of Macrofunctions & Primitives
  - Altera Library
    - Maxplus2
    - Megacore
- Can Be Used to Control Physical Implementation of Design Within the PLD
- Vendor-specific Libraries Improve Performance & Efficiency of Designs
- Altera Provides a Collection of Library of Parameterized Modules (LPMs) Plus Other Megafunctions and Primitives

130

Library Altera/LPM

- **Library** Altera **;**
  - Contains the Following Packages:
    - **Maxplus2** (Component Declarations for All Primitives and Old-style Megafunction Altera Libraries)
    - **Megacore** (Component Declarations for Some Altera Megacores)
- **Library** LPM**;**
  - Contains the Following Packages:
    - **Lpm_components** (Component Declarations for All Altera LPM Functions)

⇨ Note: See MAX+PLUS II or Quartus II Online Help for More Information

131

LPMs

- **L**ibrary of **P**arametrized **M**odules
  - Large Building Blocks That Are Easily Configurable By Using the MegaWizard Plug-In Manager
- Altera's LPMs Have Been Optimized to Access the Architectural Features of Altera Devices

132

LPM Instantiation

- All of the Altera LPM Macrofunctions Are Declared in the Package **lpm_components.all** in the **LIBRARY lpm;**
- The **MegaWizard Plug-in Manager** in Quartus II and MAX+plus II Creates the VHDL Code Instantiating the LPM or Megafunction
- After the Code Is Created You Will See at Top of the VHDL Code:

```
LIBRARY lpm;
Use lpm.lpm_components.all;
```


133

Manual LPM Instantiation – LPM_MUX

• Quartus II or MAX+plus II Online HELP: VHDL Component Declaration:

```
COMPONENT lpm_mux
  GENERIC (LPM_WIDTH: POSITIVE;
    LPM_WIDTHS: POSITIVE;
    LPM_PIPELINE: INTEGER := 0;
    LPM_SIZE: POSITIVE;
    LPM_HINT: STRING := UNUSED);
  PORT (data: IN STD_LOGIC_2D(LPM_SIZE-1 DOWNTO 0, LPM_WIDTH-1 DOWNTO 0);
    aclr: IN STD_LOGIC := '0';
    clock: IN STD_LOGIC := '0';
    sel: IN STD_LOGIC_VECTOR(LPM_WIDTHS-1 DOWNTO 0);
    result: OUT STD_LOGIC_VECTOR(LPM_WIDTH-1 DOWNTO 0));
END COMPONENT;
```

```
LIBRARY IEEE;
USE IEEE.std_logic_1164.all;
USE IEEE.std_logic_arith.all;
USE IEEE.std_logic_signed.all;

LIBRARY lpm;
USE lpm.lpm_components.all;

ENTITY tst_mux IS
PORT (a : in std_logic_2d (3 downto 0, 15 downto 0);
                sel : in std_logic_vector(1 downto 0);
                y : out std_logic_vector (15 downto 0));
END tst_mux;

ARCHITECTURE behavior OF tst_mux IS
BEGIN

u1: lpm_mux GENERIC MAP(lpm_width => 16, lpm_size => 4, lpm_widths => 2)
            PORT MAP (data => a, sel => sel,  result => y);

END behavior;
```


134

Manual LPM Instantiation – LPM_MULT

```
LIBRARY IEEE;
USE IEEE.std_logic_1164.all;
USE IEEE.std_logic_unsigned.all;

LIBRARY lpm;
USE lpm.lpm_components.all;

ENTITY tst_mult IS
PORT ( a, b : in std_logic_vector(7 downto 0);
         q_out  : out std_logic_vector(15 downto 0));
END tst_mult;

ARCHITECTURE behavior OF tst_mult IS

BEGIN

      u1 : lpm_mult GENERIC MAP (lpm_widtha => 8, lpm_widthb => 8,
                                 lpm_widths => 16, lpm_widthp => 16)
                    PORT MAP(dataa => a, datab => b, result => q_out);

END behavior;
```


135

Accessing the MegaWizard

- Altera's IP, Megafunction, & LPMs Accessed and Edited through the MegaWizard Plug-In Manager



136

MegaWizard Files

- The MegaWizard Plug-In Manager Produces Three Files Relevant to VHDL
  - my_ram.vhd
    - Instantiation and parameterisation of Megafunction
  - my_ram.cmp
  - Component declaration for use in higher level file
  - my_ram_inst.vhd
    - Instantiation of my_ram for use in higher level file

```
component my_ram
    PORT
    (
        data        : IN STD_LOGIC_VECTOR (7 DOWNTO 0);
        wraddress       : IN STD_LOGIC_VECTOR (4 DOWNTO 0);
        rdaddress       : IN STD_LOGIC_VECTOR (4 DOWNTO 0);
        wren        : IN STD_LOGIC  := '1';
        wrclock     : IN STD_LOGIC ;
        rdclock     : IN STD_LOGIC ;
        q       : OUT STD_LOGIC_VECTOR (7 DOWNTO 0)
    );
end component;
```

```
my_ram_inst : my_ram PORT MAP (
        data    => data_sig,
        wraddress   => wraddress_sig,
        rdaddress   => rdaddress_sig,
        wren    => wren_sig,
        wrclock  => wrclock_sig,
        rdclock  => rdclock_sig,
        q    => q_sig
    );
```


137

Benefits of LPMs

- Industry Standard
- Larger Building Blocks, So You Don't Have to Start From Scratch
  - Reduces Design Time
  - Therefore, Faster Time-to-market
- Easy to Change the Functionality by Using the MegaWizard
- Consistent Synthesis

138

Exercise 6

Please go Exercise 6

Altera Technical Support

- Reference Quartus II On-Line Help
- Consult Altera Applications (Factory Applications Engineers)
  - Hotline: (800) 800-EPLD (7:00 a.m. - 5:00 p.m. PST)
  - MySupport: http://www.altera.com/mysupport
- Field Applications Engineers: Contact Your Local Altera Sales Office
- Receive Literature by Mail: (888) 3-ALTERA
- FTP: ftp.altera.com
- World-Wide Web: http://www.altera.com
  - Use Solutions to Search for Answers to Technical Problems
  - View Design Examples

Learn More Through Technical Training

Instructor-Led Training	On-Line Training
	
With Altera's instructor-led training courses, you can: ➢Listen to a lecture from an Altera technical training engineer (instructor) ➢Complete hands-on exercises with guidance from an Altera instructor ➢Ask questions and receive real-time answers from an Altera instructor ➢Each instructor-led class is one day in length (8 working hours).	With Altera's on-line training courses, you can: ➢Take a course at any time that is convenient for you ➢Take a course from the comfort of your home or office (no need to travel as with instructor-led courses) ➢Each on-line course will take approximately 2-3 hours to complete.

www.altera.com/training

View Training Class Schedule & Register for a Class

Appendix

ATTRIBUTES

<signal_name> : IN STD_LOGIC_VECTOR(7 DOWNTO 0)

- **'HIGH** - 7
- **'LOW** - 0
- **'RIGHT** - 0
- **'LEFT** - 7
- **'RANGE** - 7 DOWNTO 0
- **'REVERSE RANGE** - 0 TO 7
- **'LENGTH** - 8

Subprograms

- Functions
- Procedures

Subprograms



145

Functions

- Format:

```
Function <Function_name> (<Input_parameters>)
Return <DATA_TYPE> Is
        {Any Declarations}
Begin
        {Functionality}
    Return <Name_of_a_declaration>
End <Function_name>;
```


146

Functions

- For Functions:
  - Only Allowable Mode for Parameters Is **in**
  - Only Allowed Object Classes Are **Constant** or **Signal**
  - If the Object Class Is Not Specified, **Constant** Is Assumed

147

Procedures

- Format:

```
Procedure <Procedure_name> (<Mode_parameters>)
    Begin
        {Functionality}
    End <Procedure_name>;
```


Procedures

- For Procedures:
  - Allowable Modes for Parameters Are **In, Out,** and **Inout**
  - Allowable Object Classes for Parameters Are **Constant, Variable** and **Signal**
  - If the Mode Is **in** and No Object Class Is Specified, Then **Constant** Is Assumed
  - If the Mode Is **Inout** or **Out** and If No Object Class Is Specified, Then **Variable** Is Assumed

Signal Assignment Inside a Process - Delay

- Δ Delta Cycle has 2 Phases:
  - Process Execution
  - Signal Update

```
LIBRARY IEEE;
USE IEEE.std_logic_1164.all;
ENTITY simp_prc IS
PORT(a, b : IN STD_LOGIC;
          y: OUT STD_LOGIC);
END simp_prc;
ARCHITECTURE logic OF simp_prc IS
SIGNAL c: STD_LOGIC;

BEGIN
PROCESS(a, b)
          BEGIN
          c <= a and b;
          y <= c;
END PROCESS;
END logic;
```



- **y** Does Not Get the Newest Value of **c** Until a Simulation Cycle Later
- Δ Delta Cycle is Non-Visible Delay (Very Small, Close to Zero)


2 Processes Vs. 1 Process
process1: PROCESS(a, b)
BEGIN
c <= a and b;
END PROCESS process1;
process2: PROCESS(c)
BEGIN
y <= c;
END PROCESS process2;
PROCESS(a, b)
BEGIN
c <= a and b;
y <= c;
END PROCESS;
a = 1 b = 1
c updated (c=1)
a,b changes a = 0 b = 1
c updated (c=0)
a,b changes a = 1 b = 1
y updated (y=1)
y updated (y=0)
c executed
y executed
simulation cycle1 (visible delay)
simulation cycle2 (visible delay)
c updated (c=1)
c updated (c=0)
y updated (y=X)
y updated (y=1)
a = 1, b = 1
a,b changes a = 0, b = 1
a,b changes a = 0, b = 1
c and y executed
• c and y Gets Executed and Updated Within the Same Simulation Cycle
• y Does Not Get the Newest Value of c Until a Simulation Cycle Later
Copyright © 2004 Altera Corporation
151


Variable Assignment - No Delay
• Δ Delta Cycle has 2 Phases:
– Process Execution
– Signal Update
LIBRARY IEEE;
USE IEEE.std_logic_1164.all;
ENTITY var IS
PORT (a, b : IN STD_LOGIC;
y : OUT STD_LOGIC);
END var;
ARCHITECTURE logic OF var IS
BEGIN
PROCESS (a, b)
VARIABLE c : STD_LOGIC;
BEGIN
c := a AND b;
y <= c;
END PROCESS;
END logic;
a = 1, b = 1
c executed and updated (c=1)
y updated (y=1)
y updated (y=0)
a,b changes a = 0, b = 1
a,b changes a = 1, b = 1
c executed and updated (c=0)
c executed and updated (c=1)
y executed
simulation cycle1 (visible delay)
simulation cycle2 (visible delay)
• c and y Gets Executed and Updated Within the Same Simulation Cycle (at the End of the Process)
• Δ Delta Cycle is Non-Visible Delay (Very Small, Close to Zero)
Copyright © 2004 Altera Corporation
152


2 Processes Vs. 1 Process
process1: PROCESS(a, b)
BEGIN
c <= a and b;
END PROCESS process1;
process2: PROCESS(c)
BEGIN
y <= c;
END PROCESS process2;
PROCESS(a, b)
BEGIN
c <= a and b;
y <= c;
END PROCESS;
a = 1 b = 1
c updated (c=1)
a,b changes a = 0 b = 1
c updated (c=0)
a,b changes a = 1 b = 1
y updated (y=1)
y updated (y=0)
c executed
y executed
simulation cycle1 (visible delay)
simulation cycle2 (visible delay)
c updated (c=1)
c updated (c=0)
y updated (y=X)
y updated (y=1)
a = 1, b = 1
a,b changes a = 0, b = 1
a,b changes a = 0, b = 1
c and y executed
• c and y Gets Executed and Updated Within the Same Simulation Cycle
• y Does Not Get the Newest Value of c Until a Simulation Cycle Later
Copyright © 2004 Altera Corporation
153